Paper 1062-2017

Data Visualization from SAS® to Microsoft SharePoint

Xiaogang (Isaac) Tang, Wyndham Destination Network

ABSTRACT

SharePoint is a popular web application framework and platform developed by Microsoft, widely used for content and document management by large companies and organizations. Connecting SAS with SharePoint combines the power of these two into one. As a continuation of my SAS Paper 11520-2016 "Releasing the Power of SAS® into Microsoft SharePoint", this paper expands on how to implement data visualization from SAS to Microsoft SharePoint. The paper shows users how to use SAS Procedures and SAS Output Delivery System (ODS) to create and send visualization output files from SAS to SharePoint Document Library. Several SAS code examples and output screenshots are included to show how to create tables, charts, plots and maps from SAS to SharePoint.

INTRODUCTION

SharePoint is a popular web application framework and platform developed by Microsoft, widely used by large companies and organizations. SAS is highly proficient for data processing and analysis. SharePoint is convenient to share content and manage user permissions. Connecting SAS with SharePoint combines the power of these two into one.

In 2009, a SAS Global Forum Paper titled "Integrating SAS® Business Intelligence with Microsoft SharePoint" described a variety of approaches to integrate the two technologies.

My SAS paper 11520-2016 "Releasing the Power of SAS® into Microsoft SharePoint" presents a simple and friendly solution to send SAS contents to SharePoint via an existing email system. SharePoint Document Library can easily be configured to assign a user-defined corporate email address. Sending PDF reports and Excel files from a SAS server to user email addresses through a corporate email system is a standard practice. By combining these two techniques, users can write a SAS program to send emails with SAS contents to a specific SharePoint Document Library with user-defined corporate email address.

As a continuation of "Releasing the Power of SAS® into Microsoft SharePoint", this paper expands on how to implement data visualization from SAS to Microsoft SharePoint.

Firstly, this paper will review how to configure SharePoint to accept contents from SAS.

Secondly, the paper will present 7 example SAS codes and output screenshots to illustrate how to create tables, charts, plots and maps from SAS to SharePoint.

- Example 1: How to display a single SAS table on SharePoint
- Example 2: How to display multiple SAS tables
- Example 3: How to display a single SAS table with sortable columns on a web page
- Example 4: How to display a single SAS plot on SharePoint
- Example 5: How to display multiple SAS plots
- Example 6: How to display a SAS map with data view on SharePoint
- Example 7: How to display a SAS plot with a sortable SAS table

Example 1, 2 and 3 show how to display SAS tables in a web page on SharePoint. Example 4,5 and 6 show how to display SAS plots in a web page on SharePoint. Example 7 shows how to display both SAS table and SAS plot in a web page on SharePoint.

CONFIGURE SHAREPOINT TO ACCEPT CONTENTS FROM SAS

Here are the steps to configure SharePoint:

1. Create a SharePoint Document Library. Please check with your SharePoint administrator to make sure you have the right permissions to do so. You can also use an existing SharePoint Document Library.
2. Open the SharePoint Document Library. Go to “Library Tools” Tab - “Library”, click the “Library Settings” button. In the “Document Library Settings”, under “Communications” you can find “Incoming e-mail settings”. Open it. Figure 1 is the screenshot for “Incoming e-mail settings”.
3. You may follow this example to set up your library. You need to assign an email address to the library, so SAS can send email to SharePoint. In this mock-up example, rm_viz@mycompany.com is the email address.

Tip: you may send some test emails with attachments from Outlook to the email address (rm_viz@mycompany.com) to make sure it’s properly set up, before we move on to the SAS code part.



Figure 1. Screenshot for SharePoint Library – Incoming e-mail settings

You can send a variety of file formats (PDF, Excel, CSV, DBF, HTML, JPEG, GIF, etc.) as attachments to SharePoint. Please note that SharePoint does have a size limit (for example, 5MB) for incoming email attachments. Please check with your SharePoint admin to find out the actual limit.

DATA VISUALIZATION TOPIC 1: DISPLAYING SAS TABLE

Assume you have successfully sent email from your Outlook to the SharePoint Document Library via the email address rm_viz@mycompany.com, and let’s also assume your own email address is myname@mycompany.com. let’s continue to the SAS coding part.

All these examples are using tables from sashelp library. Sashelp is a native SAS library. SAS provides over 200 data sets in the sashelp library. For more information, you may check out the official SAS documentation “SAS Help Data Sets”.

CODE EXAMPLE 1 – HOW TO DISPLAY A SINGLE SAS TABLE ON SHAREPOINT

Data visualization starts with table. Table view might seem plain, but it is still very important to show the numbers. This example shows how to display a SAS table in a web page on SharePoint.

The SAS table is Sashelp.rockpit. Output web page is viz_onetable.html.

SharePoint Output Screenshot:



Obs	DATE	AMOUNT
1	01JAN1998	-84000
2	01JAN1999	-36000
3	01JAN2000	-36000
4	01JAN2001	-120000
5	01JAN2002	-36000
6	01JAN2003	-26000

Figure 2. Screenshot to display a SAS table on SharePoint

SAS Code:

```
ods html body="/user/&sysuserid./viz_onetable.html" style=festival;
title1 h = 11pt "SAS Table sashelp.rockpit";
PROC print DATA=Sashelp.rockpit; RUN;
ods html close;

filename outbox email 'nul';
data _null_;
file outbox
to=("rm_viz@mycompany.com","myname@mycompany.com")
from=("myname@mycompany.com")
attach= "/user/&sysuserid./viz_onetable.html"
subject="viz one table"; run;
```

The SharePoint email address rm_viz@mycompany.com needs to be in the **To** line. In this case, SAS will send the web page to the SharePoint Document Library and to the user email box.

The **From** line is equally important. Please use the same email address when you send test emails from Outlook to SharePoint. In this case, it's myname@mycompany.com. SharePoint server needs to recognize this email address within its email directory otherwise it will reject the email.

The **Attach** line includes the path of the SAS file in the SAS server.

Since we choose not to save original email, the **Subject** line is only for reference use. The email content text can be blank.

Please don't expect the file to arrive in the SharePoint Document Library instantly after SAS code runs. Allow 1 minute or so before refreshing the Library.

CODE EXAMPLE 2 – HOW TO DISPLAY MULTIPLE SAS TABLES

This example shows how to display two SAS tables in a web page on SharePoint.

The two SAS tables are Sashelp.rockpit and Sashelp.citiyr. Output web page is viz_twotables.html.

SharePoint Output Screenshot:



Obs	DATE	AMOUNT
1	01JAN1998	-84000
2	01JAN1999	-36000
3	01JAN2000	-36000
4	01JAN2001	-120000
5	01JAN2002	-36000
6	01JAN2003	-26000

Obs	DATE	PAN	PAN17	PAN18	PANF	PANM
1	1980	227757	172456	138358	116869	110888
2	1981	230138	175017	140618	118074	112064
3	1982	232520	177346	142740	119275	113245
4	1983	234799	179480	144591	120414	114385
5	1984	237001	181514	146257	121507	115494
6	1985	239279	183583	147759	122631	116648
7	1986	241625	185766	149149	123795	117830
8	1987	243942	187988	150542	124945	118997
9	1988	246307	189867	152113	126118	120189
10	1989	248762	191570	153695	127317	121445

Figure 3. Screenshot to display multiple SAS tables on SharePoint

SAS Code:

```
ods html body="/user/&sysuserid./viz_twotables.html" style=festival;

title1 h = 11pt "SAS Table sashelp.rockpit";
PROC print DATA=Sashelp.rockpit; RUN;

title1 h = 11pt "SAS Table Sashelp.citiyr";
PROC print DATA=Sashelp.citiyr; RUN;

ods html close;

filename outbox email 'nul';
data _null_;
file outbox
to=("rm_viz@mycompany.com","myname@mycompany.com")
from=("myname@mycompany.com")
attach= "/user/&sysuserid./viz_twotables.html"
subject="viz two tables"; run;
```

CODE EXAMPLE 3 – HOW TO DISPLAY A SINGLE TABLE WITH SORTABLE COLUMNS

This example illustrates how to display a table web page with sortable columns, which has a great advantage over static table view. You will need to upload the two JavaScript files to the same SharePoint Document Library (**jquery.min.js** and **jquery.tablesorter.min.js**). You may also directly get these two files online (http://goo.gl/Pg0GB and http://goo.gl/ruKEb). Please see Reference 6 for more information.

Sashelp.class is the SAS table. Output web page is viz_sortabletable.html.

SharePoint Output Screenshot:



lytics/Visualization/viz_sortabletable.html

SAS Table sashelp.class
Click on any column header to sort the table.

Obs	Name	Sex	Age	Height	Weight
1	Alfred	M	14	69.0	112.5
2	Alice	F	13	56.5	84.0
3	Barbara	F	13	65.3	98.0
4	Carol	F	14	62.8	102.5
5	Henry	M	14	63.5	102.5
6	James	M	12	57.3	83.0
7	Jane	F	12	59.8	84.5
8	Janet	F	15	62.5	112.5
9	Jeffrey	M	13	62.5	84.0
10	John	M	12	59.0	99.5
11	Joyce	F	11	51.3	50.5
12	Judy	F	14	64.3	90.0
13	Louise	F	12	56.3	77.0

ytics/Visualization/viz_sortabletable.html

SAS Table sashelp.class
Click on any column header to sort the table.

Obs	Name	Sex	Age	Height	Weight
1	Alfred	M	14	69.0	112.5
5	Henry	M	14	63.5	102.5
6	James	M	12	57.3	83.0
9	Jeffrey	M	13	62.5	84.0
10	John	M	12	59.0	99.5
15	Philip	M	16	72.0	150.0
16	Robert	M	12	64.8	128.0
17	Ronald	M	15	67.0	133.0
18	Thomas	M	11	57.5	85.0
19	William	M	15	66.5	112.0
2	Alice	F	13	56.5	84.0
3	Barbara	F	13	65.3	98.0
4	Carol	F	14	62.8	102.5

Figure 4. Screenshot to display a SAS table with sortable columns on SharePoint. The left shows table in original order. The right shows table sorted by column "Sex".

SAS Code:

```
%macro ods_html_sort_table;
<script src='./jquery.min.js'></script>
<script src='./jquery.tablesorter.min.js'></script>
<script>$(document).ready(function(){$('.table').tablesorter({widgets:
['zebra']});});</script>
%mend;
ods html body="/user/&sysuserid./viz_sortabletable.html"
headtext="%ods_html_sort_table" style=festival;
title1 h = 11pt "SAS Table sashelp.class";
title2 h = 11pt "Click on any column header to sort the table.";
PROC print DATA=Sashelp.class; RUN;
ods html close;

filename outbox email 'nul';
data _null_;
file outbox
to=("rm_viz@mycompany.com","myname@mycompany.com")
from=("myname@mycompany.com")
attach= "/user/&sysuserid./viz_sortabletable.html"
subject="viz sortable table"; run;
```

DATA VISUALIZATION TOPIC 2: DISPLAYING SAS PLOTS AND MAPS

CODE EXAMPLE 4 – HOW TO DISPLAY A SINGLE SAS PLOT

This example uses PROC SGPLOT to create a bar chart to show frequency by cylinders. Table source is Sashelp.cars. The output web page is viz_oneplot.html. The bar chart is created in SAS as a jpeg file named “barchart1.jpeg”.

SharePoint Output Screenshot:



Figure 5. Screenshot to display a SAS plot on SharePoint

SAS Code:

```
/*For the purpose of line 1, please see reference No.7 */
ods tagsets.sasreport13(id=EGSR) gtitle gfootnote;

ODS GRAPHICS / RESET IMAGENAME = 'barchart' IMAGEFMT =JPEG;

ods html path  = "/user/&sysuserid./"
GPATH="/user/&sysuserid./" (url="./")
file="viz_oneplot.html" style=festival;
PROC SGPLOT DATA = sashelp.cars;
 VBAR Cylinders;
 TITLE1 'sashelp.cars Summary by Cylinders'; RUN;
ods html close;

filename outbox email 'nul';
data _null_;
file outbox
to=("rm_viz@mycompany.com","myname@mycompany.com")
from=("myname@mycompany.com")
attach= ("/user/&sysuserid./viz_oneplot.html"
"/user/&sysuserid./barchart1.jpeg")
subject="viz one plot";run;
```

CODE EXAMPLE 5 – HOW TO DISPLAY TWO PLOTS

This example uses PROC SGPLOT to create two charts. The first is a bar chart to show frequency by cylinders, table source is Sashelp.cars. The second chart is a box plot to show cholesterol distribution by weight class. Table source is Sashelp.heart. The output web page is viz_twoplots.html. The bar chart is created in SAS as a jpeg named “twoplot1.jpeg”. The box plot is created in SAS as a jpeg named “twoplot3.jpeg”. Please note that, if we add a third plot, the name will be “twoplot5.jpeg”. It’s important to use exactly the same file names when sending these SAS output files as attachments to SharePoint.

SharePoint Output Screenshot:



Figure 6. Screenshot to display two SAS plots on SharePoint

SAS Code:

```
ods tagsets.sasreport13(id=EGSR) gtitle gfootnote;

ODS GRAPHICS / RESET IMAGENAME = 'twoplot' IMAGEFMT =JPEG;

ods html path  = "/user/&sysuserid./"
GPATH="/user/&sysuserid./" (url="./")
file="viz_twoplots.html" style=festival;

PROC SGPLOT DATA = sashelp.cars;
 VBAR Cylinders;
 TITLE1 'sashelp.cars Summary by Cylinders';
RUN;

proc sgplot data=sashelp.heart;
  title "Cholesterol Distribution by Weight Class";
  hbox cholesterol / category=weight_status;
run;

ods html close;

filename outbox email 'nul';
data _null_;
file outbox
to=("rm_viz@mycompany.com","myname@mycompany.com")
from=("myname@mycompany.com")
attach= ("/user/&sysuserid./viz_twoplots.html"
"/user/&sysuserid./twoplot1.jpeg" "/user/&sysuserid./twoplot3.jpeg")
subject="viz two plots";
run;
```

CODE EXAMPLE 6 – HOW TO DISPLAY A MAP WITH DATA VIEW

This example largely is quoting a SAS.com Sample Code 53367: “Create a map with the SGPLOT procedure.” Please refer to this web page for more details: http://support.sas.com/kb/53/367.html.

The sample code uses the SAS® 9.4 POLYGON statement in PROC SGPLOT to create a map using the SAS/GRAPH® MAPSGFK.US map data set. SAS/GRAPH software is required to run this sample.

The output web page is us_map.html. The map is created in SAS as a jpeg file named “Map1.jpeg”.

SharePoint Output Screenshot:



Figure 7. Screenshot to display a SAS map with data view on SharePoint

SAS Code:

```
data us;
   set mapsgfk.us;
   where id ne 'US-11'; run;

data response;
   length state 5;
   input state count;
   datalines;
37 100
45 200
6 300
8 400
```

```
48 600
30 500
17 300
run;

proc sort data=response;
   by state;
run;

/* Combine the response data with the map data */
data all;
   merge us response;
   by state;
   id = catx('-', id, segment);
run;

/* Define a format for the response data */
proc format;
   value mapfmt
        .='No Data'
      low-100='100 and under'
      101-200='Between 100 and 200'
      201-300='Between 200 and 300'
      301-400='Between 300 and 400'
      401-high='Over 400';
run;

/* Define an attribute map for the response data */
data attrmap;
   id = 'maparea';
   textcolor='black';
   input value $20. @22 fillcolor $;
   datalines;
No Data              beige
100 and under        cx74c476
Between 100 and 200  cx006d2c
Between 200 and 300  cx756bb1
Between 300 and 400  cx41b6c4
Over 400             cx253494
;
run;

/* Make the SAS/GRAPH annotate macros available */
%annomac;
%centroid(us,centers,id);

/* Define a label to be placed at the center of each polygon */
data all;
   set all centers(rename=(x=xcen y=ycen) in=a);
   if a then label=fipstate(substr(id,4,2));
  /* Adjust a few label coordinates */
   if label = 'ID' then ycen + -.025;
```

```
   if label = 'MI' then ycen + -.025;
   if label = 'HI' then ycen + -.01;
   if label = 'NH' then ycen + -.012;
   if label = 'VT' then ycen + .01;
   if label = 'MD' then ycen + .007;
   if label = 'AK' then ycen + .01;
   if label = 'DE' then do;
      ycen + -.005;
      xcen + .005;
   end;
run;
/* The IMAGEMAP option enables you to view the data tips for the graph
*/
ods tagsets.sasreport13(id=EGSR) gtitle gfootnote;
ods graphics / reset width=800px height=600px imagefmt=png
imagename='Map' imagemap=on tipmax=4000;

ods html path  = "/user/&sysuserid./"
GPATH="/user/&sysuserid./" (url="./")
file="us_map.html" style=festival;

title 'Counts by State';
proc sgplot data=all dattrmap=attrmap ;
   format count mapfmt.;
  /* Draw each polygon */
   polygon x=x y=y id=id / group=count attrid=maparea
           outline lineattrs=(color=gray99)
           fill fillattrs=(transparency=0.5)
           dataSkin=matte name='poly'
        /* Remove the TIP= option to prevent data tips */
           tip=(statecode count);
  /* Label each polygon with the LABEL variable value */
   scatter x=xcen y=ycen / markerchar=label
        /* Remove the TIP= option to prevent data tips */
           tip=(id);
   keylegend 'poly' / title='Count Value: ';
   xaxis offsetmin=0.01 offsetmax=0 display=none;
   yaxis offsetmin=0.01 offsetmax=0 display=none;
run;
ods html close;

filename outbox email 'nul';
data _null_;
file outbox
to=("rm_viz@mycompany.com","myname@mycompany.com")
from=("myname@mycompany.com")
attach=("/user/&sysuserid./us_map.html" "/user/&sysuserid./Map1.png")
subject="SAS Map example";
run;
```

CODE EXAMPLE 7 – HOW TO DISPLAY A SORTABLE TABLE AND A PLOT

This example is to illustrate that a SAS table and a SAS plot can be displayed in one page. The SAS table has sortable columns.

You will need to upload the two JavaScript files to the same SharePoint Document Library (jquery.min.js and jquery.tablesorter.min.js). You may also directly get these two files online (http://goo.gl/Pg0GB and http://goo.gl/ruKEb). Please see Reference 6 for more information.

The output web page is viz_table_plot.html. SAS table sashelp.class is displayed as a table with sortable columns. This example uses PROC SGPLOT to create a bar chart to show frequency by cylinders. Table source is Sashelp.cars. The output web page is viz_oneplot.html. The bar chart is created in SAS as a jpeg named “barchart1.jpeg”.

SharePoint Output Screenshot:



Figure 8. Screenshot to display a SAS plot and a SAS table with sortable columns

SAS Code:

```
%macro ods_html_sort_table;
<script src='./jquery.min.js'></script>
<script src='./jquery.tablesorter.min.js'></script>
<script>$(document).ready(function(){$('.table').tablesorter({widgets:
['zebra']});});</script>
%mend;

ods tagsets.sasreport13(id=EGSR) gtitle gfootnote;
ODS GRAPHICS / RESET IMAGENAME = 'barchart' IMAGEFMT =JPEG;
```

```
ods html path  = "/user/&sysuserid./"
GPATH="/user/&sysuserid./" (url="./")
file="viz_table_plot.html"
headtext="%ods_html_sort_table" style=festival;

PROC SGPLOT DATA = sashelp.cars;
 VBAR Cylinders;
 TITLE1 'sashelp.cars Summary by Cylinders';
RUN;

title1 h = 11pt "SAS Table sashelp.class";
title2 h = 11pt "Click on any column header to sort the table.";

PROC print DATA=Sashelp.class; RUN;
ods html close;

filename outbox email 'nul';
data _null_;
file outbox
to=("rm_viz@mycompany.com","myname@mycompany.com")
from=("myname@mycompany.com")
attach= ("/user/&sysuserid./viz_table_plot.html"
"/user/&sysuserid./barchart1.jpeg")
subject="viz plot and sortable table";
run;
```

CONCLUSION

The whole process of data visualization from SAS® to Microsoft SharePoint can serve as a practical low-cost Business Intelligence toolbox. It only requires SAS standard license and Microsoft SharePoint standard license. It can build up KPI dashboards on SharePoint, without the need to buy additional Analytics products, which could be costly.

Both SharePoint and SAS are widely popular in Enterprise environment. The approach in this paper is straight forward and easily applicable.

With appropriate and creative implementations, it can save company software costs and provide business insights and guidance with highly efficient business reporting and meaningful data visualization.

One thing to note is that creating such data visualization will require multidimensional skills (SAS, SharePoint, HTML) from a business analyst. The analyst will also need guidance and user feedback from business stakeholders to create the best solution.

It's highly recommended to try out this new approach with the example codes. The concept is full of potential. The key is to implement it practically in your company. Once the process is implemented, you will enjoy a great journey with it.

REFERENCES

1. Herderson, D. R. and S. Alexandre 2009. “Integrating SAS® Business Intelligence with Microsoft.” *Proceedings of SAS Global Forum 2009.*

2. Slaughter, S. J. and L.D. Delwiche 2010. “Using PROC SGPLOT for Quick High-Quality Graphs.” *Proceedings of SAS Global Forum 2010.*

3. Slaughter, S. J. and L.D. Delwiche 2011. “Graphing Made Easy with SG Procedures.” *Proceedings of SAS Global Forum 2011.*

4. Xiaogang Tang 2016. “Releasing the Power of SAS® into Microsoft SharePoint.” *Proceedings of SAS Global Forum 2016.*

5. SAS.com “Sample 53367: Create a map with the SGPLOT procedure.” Accessed March 1, 2017. http://support.sas.com/kb/53/367.html

6. Heuristic Andrew, “Easy sortable HTML tables in SAS ODS”. Accessed March 1, 2017. https://heuristically.wordpress.com/2013/01/17/sortable-html-tables-sas-ods/

7. Chris Hemedinger, “embed titles in png with ods graphics”. Accessed March 1, 2017. https://communities.sas.com/t5/SAS-GRAPH-and-ODS-Graphics/embed-titles-in-png-with-ods-graphics/m-p/160980#M6034

RECOMMENDED READING

- *Base SAS® Procedures Guide*
- *SAS Graphics Samples Output Gallery https://support.sas.com/sassamples/graphgallery/*
- *PROC GPLOT Graphics Samples Output Gallery https://support.sas.com/sassamples/graphgallery/PROC_GPLOT.html*
- *PROC SGPLOT Graphics Samples Output Gallery https://support.sas.com/sassamples/graphgallery/PROC_SGPLOT.html*
- *PROC GCHART Graphics Samples Output Gallery https://support.sas.com/sassamples/graphgallery/PROC_GCHART.html*
- *PROC GMAP Graphics Samples Output Gallery https://support.sas.com/sassamples/graphgallery/PROC_GMAP.html*
- *SAS ODS (Output Delivery System) http://support.sas.com/rnd/base/ods/index.html*

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Xiaogang (Isaac) Tang
Wyndham Destination Network
xiaogang.tang@rci.com